Universidad San Francisco de Quito
Curso: Introducción a gnuplot
Física

Instructor: Julio César Andrade L.

Ejercicios

1. Ejemplos a desarrollar en la clase

Ploteo de puntos

```
punto1="<echo 0 2"
punto2="<echo -3 3"
punto3="<echo -1 2.5"
punto4="<echo 0 3.5"
punto5="<echo 1 2.2"
punto6="<echo -1 1"

plot[-4:4][0:4.5] punto1 pt 6 ps 2 lc rgb 'red',\
                  punto2 pt 3 ps 2 lc rgb 'cyan',\
                  punto3 pt 4 ps 2 lc rgb 'green',\
                  punto4 pt 2 ps 2 lc rgb 'black',\
                  punto5 pt 22 ps 2 lc rgb 'yellow',\
                  punto6 pt 24 ps 2 lc rgb 'blue'
```



Ploteo de funciones

```
set xrange [-5:5]
set yrange [-50:50]
set xlabel 'eje de las x'
set ylabel 'eje de las y'
set title "Mi primer gráfico en GNU Plot"
set key box
set key spacing 3 font "Helvetica, 8"
set key at -3.5,45
f(x)=7*x+2
g(x)=-2*x**2+3*x-3
h(x)=(x**2-3*x)/(x**2-2)
i(x)=5*sin(x)
plot f(x) lw 1.2 lc rgb 'yellow',\
     g(x) lw 1.2 lc rgb 'red',\
     h(x) lw 1.2 lc rgb 'cyan' dt 4,\
     i(x) lw 1.2 lc rgb 'blue' dt 2
```



Ejemplo práctico 1

```
punto_1="<echo 5 8"
punto_2="<echo 5 4"
f(x)= (x<5) ? -(x-5)**2+8 : 1/0
g(x)= (x>5) ? (x-5)**2+4 : 1/0
set zeroaxis
set xrange [-1:8]
set yrange [-5:15]
set label "x_{0}" at 5,-1
set label "Li" at -0.5,8
set zeroaxis lt 8 lw 2
set title "Limites laterales distintos"
set arrow 1 from 0,8 to 5,8 nohead lt 0 lw 2
set arrow 2 from 5,0 to 5,4 nohead lt 0 lw 2
set arrow 3 from 3,12 to 4.9,8.4 lt 1 lw 2
set linetype 1 lc rgb '#A3001E'
set label textcolor rgb "cyan" 'Discontinuity point' at 1.8,12.5
set label textcolor rgb "blue" 'f(x<x_0)' at 2,5
set label textcolor rgb "blue" 'f(x>x_0)' at 6.7,11
set style fill transparent solid 0.35 noborder
filter(x,min,max) = (x > min && x < max) ? x : 1/0
plot f(x) lt 1 lw 3 lc rgb 'red' notitle, \
     g(x) lt 1 lw 3 lc rgb 'blue' with filledcurves above y=0 notitle, \
     g(x) lt 1 lw 3 lc rgb 'red' notitle,\
     punto_1 pt 6 ps 2 lt 7 lw 2 notitle,punto_2 pt 7 lt 7 ps 2 notitle
```



Ejemplo práctico 2

```
reset
set term tikz standalone size 11cm, 13cm
set output 'ejemplo6.tex'

G=6.674*(10**(-11))
M=5.972*(10**(24))

m1=4.340*(10**(22))
m2=7.340*(10**(22))
m3=9*(10**(22))
m4=12.0*(10**(22))
m5=20.0*(10**(22))

V1(x)= -(G*m1*M)/x
V2(x)= -(G*m2*M)/x
V3(x)= -(G*m3*M)/x
V4(x)= -(G*m4*M)/x
V5(x)= -(G*m5*M)/x
Veff1(x)= (10**(-37))*V1(x)
Veff2(x)= (10**(-37))*V2(x)
Veff3(x)= (10**(-37))*V3(x)
Veff4(x)= (10**(-37))*V4(x)
Veff5(x)= (10**(-37))*V5(x)
set arrow from 0.5,-1.5 to 2.5,-6.5 dt 2 lc 'red'
set label '$m\uparrow$' textcolor rgb "red" at 2.5,-5.5
set label 'Energy' textcolor rgb "blue" at 4,-14.5
#set key box
set key at 9.5,-5
set title "Gravitational potential $V(r) = -\\frac{GmM}{r}$"
set xlabel 'r[m]'
set ylabel '$10^{-37} V(r)$ [Joules]'

plot[0:10][-25:0] Veff1(x) lw 2 lt 1 lc 'black' title '$m_1$',\
                  Veff2(x) lw 2 lt 2 lc 'green' title '$m_2$',\
                  Veff3(x) lw 2 lt 3 lc 'blue' title '$m_3$',\
                  Veff4(x) lw 2 lt 4 lc 'red' title '$m_4$',\
                  Veff5(x) lw 2 lt 5 lc 'yellow' title '$m_5$',\
                  -15 with lines dt 4 lt 3 lw 1 lc "blue" notitle
unset output
system("pdflatex ejemplo6")
```



Ejemplo práctico 3

```
reset
set term tikz standalone size 7cm, 7cm
set output 'ejemplo7.tex'

beta = -2.92*(10**8)
g = 0.54*(10**(-10))
xi = 1.56*(10**(9))

w(x) = (sqrt(g) / (x* (xi*(x**2) + 0.5*beta)**0.5))*(10**9)

set xlabel '$P_{0}$ $[C/m^2]$'
set ylabel '$\xi_{180^{\circ}}$ [nm]'


set label '$P_{crit} = \sqrt{-\frac{\beta}{2\zeta}}=0.3052 C/m^2$' textcolor rgb "red" at 0.25,1.0 rotate by 90
set arrow from 0.3052, graph 0 to 0.3052, graph 1 lc rgb 'red' dt 2 nohead

plot[0:0.75][0:12] w(x) lw 2 lt 1 lc 'blue' notitle

unset output
system("pdflatex ejemplo7")
```



Ejemplo práctico 4

```
reset
set term tikz standalone size 10cm, 10cm
set output 'ejemplo12.tex'

set xrange[-2:2]
set yrange[0:20]

k=25
h=1
w=2.85
E(x) = (x+0.5)*h*w

unset key
set arrow from 0,0 to 0,17 lw 2 dt 2 lc 'blue'
set label '$Energy\quad E = (n+\frac{1}{2})\hbar\omega$' at -1,18 textcolor 'blue'
set xlabel 'internuclear separation'
set arrow from 0.4,10.2 to 0.4,E(4)-0.2 lc 'red'
set label '$\hbar\omega$' at 0.43,11.5 textcolor 'red'
set arrow from 1.2,14.8 to 0.65,12 lc 'red'
set label 'Energy transition' at 0.5,15.2 textcolor 'red'
set label '$E_{p}=\frac{1}{2}k x^2$' at -1.95,14 textcolor 'black'

set label '$n=0$' at 1.2,E(0)+0.4 textcolor 'blue'
set label '$n=1$' at 1.2,E(1)+0.4 textcolor 'blue'
set label '$n=2$' at 1.2,E(2)+0.4 textcolor 'blue'
set label '$n=3$' at 1.2,E(3)+0.4 textcolor 'blue'
set label '$n=4$' at 1.2,E(4)+0.4 textcolor 'blue'

plot 0.5*k*x**2 lc rgb 'black' lw 2 notitle, E(0), E(1), E(2), E(3), E(4)

unset output
system("pdflatex ejemplo12")
```



Ejemplo práctico 5

```
reset
f(x) = a*x + b
FIT_LIMIT = 1e-6

fit f(x) 'data.txt' using 1:2 via a, b
```

```
reset

set grid
set xlabel 'x'
set ylabel 'y'
set title 'My first fitting data plot'

set key at 2.8,6

plot 'data.txt' title 'Data' lc rgb 'red', 2.63106*x + 4.99387 title 'f(x) = a x+b' lw 2
```



Ejemplo práctico 6




```
reset
f(x) = a*x**2 + b*x +c
FIT_LIMIT = 1e-6

fit f(x) 'data2.txt' using 1:2 via a, b, c

reset

set grid
set xlabel 'x'
set ylabel 'y'
set title 'My second fitting data plot'

set key at 3.5,6

plot 'data2.txt' title 'Data' lc rgb 'red',\
     3.06999*x**2 - 4.14424*x + 2.93278 title 'f(x) = a x^2 + b x + c' lw 2
```

Ejemplo práctico 7

$$f(x,y) = e^{-0.0001x}\sin(xy)\cos(x-3y)$$






```
reset
set term tikz standalone size 11.5cm, 11.5cm
set output 'ejemplo10.tex'

#set pm3d
set xrange[-2:2]
set yrange[-2:2]
set isosamples 35
set hidden3d
#set key outside
set title '$f(x,y) = e^{-0.0001x}\sin{(xy)}\cos{(x-3y)}$'
set xlabel '$x$ axis'
set ylabel '$y$ axis'
set zlabel '$z$ axis'
set contour both
unset key
splot exp(-0.0001*x)*sin(x*y)*cos(x-3*y)

unset output
system("pdflatex ejemplo10")
```

```
reset
set term tikz standalone size 10cm, 10cm
set output 'ejemplo11.tex'

set xrange[-2:2]
set yrange[-2:2]
set isosamples 50
set view map
unset surface
set hidden3d
set contour base
set cntrparam levels 10
set title 'Nivel curves for $f(x,y)$'
splot exp(-0.0001*x)*sin(x*y)*cos(x-3*y) notitle

unset output
system("pdflatex ejemplo11")
```

Ejemplo práctico 8




```
reset
set term tikz standalone size 11.5cm, 11.5cm
set output 'ejemplo15.tex'

set parametric
set xrange [-110:110]
set yrange [-110:120]
set trange [0:300]
set samples 10000

set title '$f(x,y) = (x(t),y(t))$'
set xlabel '$x$ axis'
set ylabel '$y$ axis'
unset key

plot -35*cos(0.5*t)+65*cos(-.35*t),-35*sin(0.5*t)-65*sin(-.35*t)

unset output
system("pdflatex ejemplo15")
```

Ejemplo práctico 9




```
reset
set parametric
set size ratio -1
set xrange [-1.2:1.2]
set yrange [-1.2:1.2]
set trange [0:2*pi]

# Style definitions
set border lw 1.5
set style line 1 lc rgb '#0060ad' lt 1 lw 2     # --- blue
set style line 2 lc rgb '#dd181f' lt 1 lw 2     # --- red

# Parametric functions for a circle
r = 1.0
h = r / sqrt(2.)
set arrow from 0,0 to h,h nohead ls 2
set label 'r' at 0.28,0.45 textcolor ls 2 font ",16"

unset key; unset tics; unset border

fx(t) = r*cos(t)
fy(t) = r*sin(t)
plot fx(t),fy(t) ls 1
```

Ejemplo práctico 10




```
reset
set parametric

set urange [0:3.0/2*pi]
set vrange [-pi/2:pi/2]

# Style definitions
set border lw 1.5
set linetype 1   lc rgb '#0060ad' lw 2     # --- blue
set linetype 2   lc rgb '#0060ad' lw 2     # --- blue
set style line 3 lc rgb '#dd181f' lw 3     # --- red

#set radius
r = 2.0
h = r / sqrt(2.)
set arrow from 0,0,0 to h,0,h front nohead ls 3
set label 'r' at 0.34,0,0.45 textcolor ls 3 font ",15"

unset key; unset tics; unset border
set isosamples 30
set hidden3d
#set contour both

# Parametric functions for the sphere
fx(v,u) = r*cos(v)*cos(u)
fy(v,u) = r*cos(v)*sin(u)
fz(v)   = r*sin(v)

splot fx(v,u),fy(v,u),fz(v)
```

2. Ejercicios de prática

Ejercicio de práctica 1

La ley de Curie-Weiss describe la susceptibilidad magnética χ de un ferromagneto en la región paramagnética sobre el punto de Curie Θ_C, o, en general, en un material casi idealmente paramagnético en el que las interacciones entre momentos magnéticos hacen que se desvíe de la ley de Curie:

$$\chi(T) = \frac{C}{(T - \Theta_C)}.$$

Hacer un plot de la Ley de Curie para un ferromagneto que posee los siguientes valores: $C = 91{,}936K$ y $\Theta_C = 278{,}5K$.

Solución:



Potencial central

Potencial efectivo para una fuerza central similar a la gravitatoria generado por dos objetos materiales de masas m_1 y m_2 viene dado por

$$V_{eff} = V_{grav}(r) + V_{cf}(r) = -\frac{Gm_1m_2}{r} + \frac{L^2}{\mu r^2}$$

donde G es la constante universal de gravitación y $\mu = (m_1m_2)/(m_1 + m_2)$ la masa efectiva. Por facilidad considere $m_1 = 1$, $m_2 = 3$, $L = 7$ y $G = 100$ y realize un gráfico ilustrativo del potencial efectivo.

Solución:



Fitting plot

En un experimento de cinemática se ha recolectado los siguientes datos:

t[seg]	x[m]
0.001	1.842
0.513	1.765
1.234	5.785
2.341	19.672
2.451	26.412
3.312	43.987
3.555	54.671
4.101	70.011
4.212	77.621
5.511	123.345

Encuentre las funciones $f(x) = ax^2 + bx + c$ y $g(x) = \sqrt{ax^2 + bx + c}$ que se ajustan a estos datos mediante gnuplot y graficarlas.

Solución:



3D plots

Realize la gráfica de las funciones $f(x, y) = \sqrt{100 - x^2 - y^2}$ y $g(x, y) = x - y$ y una gráfica de sus contornos.





3. Soluciones de los ejercicios

Ejercicio 1

```
reset
set term tikz standalone size 11cm, 13cm
set output 'ejercicio1.tex'

C = 91.936
tc= 278.5

set arrow from 278.5,0 to 278.5,17 dt 2 lc 'red' nohead
f(x) = C/abs(x-tc)

g(x) = (x>278.5) ? 1/f(x) : 1/0

set xlabel 'T[K]'
set ylabel '$\chi$'
set label '$\Theta_{C}$' at 260.5,-0.5 textcolor rgb 'red'
set label '$1/\chi$' at 350, 4 textcolor 'red'
set label '\textbf{Curie-Weiss law}' at 105,11
set label '$\chi = \frac{C}{(T-\Theta_{C})}$' at 115,10

unset key
set title 'Ferromagnetic susceptibility'
plot[0:410][0:17] f(x) lw 2.5 lc rgb 'blue', g(x) dt 2 lc 'red'

unset output
system("pdflatex ejercicio1")
```

Ejercicio 2

```
reset
set term tikz standalone size 10cm, 10cm
set output 'ejercicio2.tex'

m1=1
m2=3
l=7
G=100

f(x)= -(G*m1*m2)/x + l**2/(0.75*x**2)
g(x)= -(G*m1*m2)/x
h(x)= l**2/(0.75*x**2)

unset key
set xlabel '$r$'
set ylabel 'energy'
set title 'Central potential'
set arrow from 3.9,199 to 0.3,-210 dt 2 lc 'black'
set label '$V_{eff}(r)$' at 4,200 textcolor 'black'
set label '$V_{g}(r)$' at 1.5,-300 textcolor 'blue'
set label '$V_{cf}(r)$' at 1.0,200 textcolor 'blue'
set label '$E>0$' at 6.0,130 textcolor 'red'
set label '$E<0$' at 6.0,-190 textcolor 'red'

plot[0:10][-500:500] f(x) lw 2 lc rgb 'cyan',\
           g(x) lw 2 lc rgb 'blue' dt 2,\
           h(x) lw 2 lc rgb 'blue' dt 2,\
           0 lw 2 lc rgb 'black' dt 2, -220 lw 2 lc rgb 'red', 100 lw 2 lc rgb 'red'


unset output
system("pdflatex ejercicio2")
```

Ejercicio 3

Primero creo un archivo de texto con los datos. Luego ejecuto el siguiente script para obtener las funciones de interpolación.

```
reset
f(x) = sqrt(a*x**2+b*x+c)
g(x) = a*x**2+b*x+c
FIT_LIMIT = 1e-6

fit f(x) 'data3.txt' using 1:2 via a, b, c
fit g(x) 'data3.txt' using 1:2 via a, b, c
```

Y luego ejecuto el siguiente script.

```
reset
set term tikz standalone size 11cm, 13cm
set output 'ejercicio3.tex'

set grid
set xlabel '$t[seg]$'
set ylabel '$x[m]$'
set title '$x = x(t)$'

set key at 3.5,120
set key spacing 1.5
f(x)=sqrt(708.048*x**2-1527.71*x+5.28453)
g(x)=4.10262*x**2+0.0577123*x+0.584621
plot 'data3.txt' title 'Data' lc rgb 'red' ps 1.5 pt 22,\
     f(x) title '$x(t) =\sqrt{a_{1}t^2+b_{1}t+c_{1}}$' lw 2 lc 'blue',\
     g(x) title '$x(t) = a_{2}t^2+b_{2}t+c_{2}$' lw 2 lc 'cyan'

unset output
system("pdflatex ejercicio3")
```

Ejercicio 4

```
reset
set term tikz standalone size 11cm, 13cm
set output 'ejercicio4.tex'
set title '\textbf{My 3d Plot}'
set key spacing 3

#set contour
set xlabel '{\bf x}'
set ylabel '{\bf y}'
set zlabel '{\bf z}'
splot 100 - x**2 - y**2 title '$f(x,y)=\sqrt{100-x^2-y^2}$' lc 'red',\
      x-y  title '$g(x,y)=x-y$' lc 'blue'

unset output
system("pdflatex ejercicio4")
```

```
reset
set term tikz standalone size 11cm, 13cm
set output 'ejercicio4a.tex'


set isosamples 50
set view map
unset surface
set hidden3d
set contour base
set cntrparam levels 10
set title '\textbf{Nivel curves for $f(x,y)$ and $g(x,y)$}'
splot 100 - x**2 - y**2 notitle,\
      x-y  notitle

unset output
system("pdflatex ejercicio4a")
```